# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=sipcrack
pkgver=0.2
pkgrel=7
groups=('blackarch' 'blackarch-cracker' 'blackarch-voip')
pkgdesc='A SIP protocol login cracker.'
license=('custom')
depends=('libpcap')
url='http://www.remote-exploit.org/codes_sipcrack.html'
arch=('x86_64' 'aarch64')
source=("http://ftp.de.debian.org/debian/pool/main/s/sipcrack/${pkgname}_${pkgver}.orig.tar.gz"
        'LICENSE')
sha512sums=('406c0f0ac4f23529e2d58f131c571492fd655d27bb466a5b972cef2c7b8ca5d550d6055ad405a4180b711a5f29027a8d47b43a912b4859ef9fccdb85069a1c45'
            '0195d97500e9a044dde046e6fbc08434d06e56b0e86c9237f4a634ef61bd864d145e9fbe445825d5e748d6c17c021ade0b4f49709c98f2b40f63aa1b96263c6f')

build() {
  cd "$pkgname-$pkgver"

  make
}

package() {
  cd "$pkgname-$pkgver"

  install -Dm 755 $pkgname "$pkgdir/usr/bin/$pkgname"
  install -Dm 755 sipdump  "$pkgdir/usr/bin/sipdump"
  install -Dm 644 "$srcdir/LICENSE" \
    "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}

